

TITLE

SUBTITLE

NAME

SUBJECT WORKINGRANGE | INSTITUTION

Zusammenfassung

In dieser wissenschaftspropädeutischen Seminararbeit aus dem Fach Informatik soll zunächst ein Modell zur Simulation eines sogenannten Magnetpendels erstellt werden. Aufgrund der Vielzahl an einwirkenden Kräften und deren Abhängigkeiten untereinander wird zur Berechnung des Modells der numerische Weg gewählt. Dazu wird ein Programm namens Simplex3 verwendet, welches entsprechende Möglichkeiten in einer Simulations- und Experimentierumgebung bereitstellt. Eine Einführung in das Programm bietet [4]. Nach der Erarbeitung des Modells folgt die Durchführung einiger Experimente, wobei der Einfluss verschiedener Größen auf die Kausalität des Magnetpendels anhand von Fraktalbildern untersucht wird.

INHALTSVERZEICHNIS

ABBILDUNGSVERZEICHNIS

PROGRAMMCODEVERZEICHNIS

HINWEIS

Dieser Arbeit liegt eine CD bei, auf der Dateien zur vergrößerten Ansicht oder digitalen Auswertung zur Verfügung stehen. Konkret befinden sich darauf folgende Dateien:

- Dieses Dokument in digitaler Form
- Sämtliche Abbildungen, die dieses Dokument enthält
- Der EDL-Code für die Durchführung der Experimente aus Abschnitt 4
- Der MDL-Code in seiner ursprünglichen Konfiguration entsprechend Fraktalbild 3 (a)

1 AUFBAU DES MAGNETPENDELS

Realität: Ein Magnetpendel ist ein Versuchsaufbau aus der Physik, der oft zur Verdeutlichung von chaotischen Systemen bzw. schwacher Kausalität verwendet wird (vgl. [1]). Hauptkomponente ist der Pendelmagnet, ein Magnet, der an einer Stange unter seinem Ankerpunkt hängt und frei schwingen kann. In geringem Abstand unter dem Pendelmagneten befindet sich der Boden des Versuchsaufbaus, auf dem einige Magnete fixiert sind. Diese sind im Bezug auf deren Polung so befestigt, dass jeder den Pendelmagneten anzieht. Ihre Anordnung folgt dabei meist einem bestimmten Muster. In der Regel werden sie in gleichmäßigem Abstand um den Mittelpunkt des Bodens angeordnet. Bei einer geraden Anzahl dieser Fixmagnete ergibt sich also eine zum Mittelpunkt des Bodens punktsymmetrische Anordnung. Die zwischen den Fixmagneten und dem Pendelmagneten auftretende resultierende Anziehungskraft muss so groß sein, dass sie den Pendelmagnet entgegen der Schwerkraft außerhalb seiner Ruheposition festhalten kann. Wird der Pendelmagnet ausgelenkt, so wird er während des Pendelvorgangs ständig von den Fixmagneten abgelenkt, bis er schließlich so viel Energie verloren hat, dass er die Anziehung eines einzelnen Magneten nicht mehr überwinden kann und er über diesem zum Stehen kommt. Dabei kann die Endposition aufgrund der Einflussnahme durch die Fixmagnete schon durch minimale Änderung der Startposition komplett verändert werden.




Abbildung 1: Skizze des Magnetpendels

Modell: Das Modell stellt ein solches Magnetpendel dreidimensional dar. Wie in Abbildung 1 zu sehen ist, liegt der Ursprung 0 des Koordinatensystems auf dem Ankerpunkt des Pen-

delmagneten P . Die Länge der Verbindungsstange zwischen 0 und P beträgt $1m$. Die beiden fixierten Magnete A und B sind auf dem theoretischen Boden des Modells, der Ebene $E : -1,05m = z$, platziert. Ausgehend vom Mittelpunkt des Bodens $M(0m|0m|-1,05m)$ ist der Fixmagnet A um $5cm$ in negative y -Richtung und der Fixmagnet B um den selben Betrag in positive y -Richtung verschoben.

Das Modell weicht in zwei Punkten von seinem realen Vorbild ab. Zunächst sind die Magnete als elektrostatische Kugeln implementiert, sodass sie eine elektrische Ladung aufweisen. Dies ist der Tatsache geschuldet, dass magnetische Anziehungskräfte sehr schwer zu berechnen sind, da Magnete Dipole sind und ein sehr unregelmäßiges Magnetfeld ausbilden, wenn sie sich gegenseitig beeinflussen. Die Anziehungskräfte zwischen geladenen Kugeln hingegen lassen sich mit den Regeln der Elektrostatik exakt und effizient berechnen. Im Zusammenhang damit ist zu beachten, dass, auch wenn es sich im Modell um elektrisch geladene Kugeln handelt, diese zugunsten des Sprachflusses weiterhin als Magnete bezeichnet werden können. Die zweite Abweichung vom realen Vorbild besteht darin, dass die Verbindungsstange des Pendelmagneten keine Dicke aufweist. Sie wird somit bei der Berechnung des Strömungswiderstandes vernachlässigt.

Jeder Magnet ist somit durch seine Ladung Q und seine Position \vec{p} definiert. Der bewegliche Pendelmagnet besitzt zusätzlich eine Geschwindigkeit \vec{v} , eine Beschleunigung \vec{a} , eine auf ihn wirkende, resultierende Kraft \vec{F}_{res} sowie eine Masse m und einen Radius r .

2 HERLEITUNG DER MODELLDYNAMIK

Die den Pendelmagneten beeinflussenden Kräfte können in zwei Kategorien unterteilt werden:

Äußere Einwirkungen: Am offensichtlichsten ist die Gesamtheit der äußeren Einwirkungen \vec{F}_{ges} , also alle Kräfte, die nicht in Zusammenhang mit der Kreisbewegung des Pendels stehen. Diese bestehen aus der Erdanziehungskraft \vec{F}_g , der zwischen dem Pendelmagneten und den festen Ablenkungsmagneten wirkenden Coulombkraft \vec{F}_c und dem Strömungswiderstand \vec{F}_{fr} . Daraus ergibt sich folgende Formel:

$$\vec{F}_{ges} = \vec{F}_g + \vec{F}_{fr} + \vec{F}_c \quad (1)$$

Diese Kraft muss dann mithilfe einer in Teilabschnitt 2.5 beschriebenen Kräftezerlegung aufgespalten werden - in ihren tangential zur Pendelbewegung wirkenden Teil \vec{F}_{tan} und den normal zur Selbigen wirkenden Teil \vec{F}_{norm} . \vec{F}_{ges} ist somit auch die Summe dieser beiden Teilkräfte.

$$\vec{F}_{ges} = \vec{F}_{tan} + \vec{F}_{norm} \quad (2)$$

Zugkraft der Verbindungsstange: Die restlichen Kräfte sind unter der Zugkraft der Verbindungsstange \vec{F}_{zug} vereint. Sie halten den Pendelmagneten in seiner Kreisbahn. Dazu zählen die Zentripetalkraft \vec{F}_{zp} , sowie die der Normalkraft \vec{F}_{norm} entgegenwirkenden Kraft \vec{F}_{geg} . Diese ist dafür verantwortlich, dass der Pendelmagnet von den magnetischen Anziehungskräften oder der Gewichtskraft nicht nach unten gezogen werden kann, was eine Verlängerung der imaginären Verbindungsstange zur Folge hätte. Die Formel für die Zugkraft der Verbindungsstange ergibt sich aus folgender Herleitung:

$$\begin{aligned} \vec{F}_{geg} &= -\vec{F}_{norm} \\ \vec{F}_{zug} &= \vec{F}_{zp} + \vec{F}_{geg} \\ \vec{F}_{zug} &= \vec{F}_{zp} - \vec{F}_{norm} \end{aligned} \quad (3)$$

2.1 Die Erdanziehungskraft

Für die Berechnung des Betrags der Erdanziehungskraft gilt folgende Formel, wobei m die Masse der Pendelkugel ist und g die Erdbeschleunigung darstellt.

$$F_g = m \cdot g$$

Die Richtung der Erdbeschleunigung entspricht der Richtung zum Erdmittelpunkt vom betrachteten Objekt aus. In Anbetracht der Größenverhältnisse zwischen Modell und Erde kann die Richtung von \vec{F}_g aber als senkrecht nach unten zeigend angenommen werden. Es ergibt sich somit folgende Formel für die Berechnung der Erdanziehungskraft im Modell.

$$\vec{F}_g = m \cdot g \cdot \begin{pmatrix} 0 \\ 0 \\ -1 \end{pmatrix} \quad (4)$$

2.2 Der Strömungswiderstand

Der Betrag des Strömungswiderstands lässt sich mit dieser Formel berechnen.

$$F_{fr} = \frac{1}{2} \cdot A \cdot c_w \cdot \rho \cdot v^2$$

Dabei ist ρ die Dichte des Mediums, v die relative Geschwindigkeit des Objekts zum Medium und A dessen maximale Querschnittsfläche in der Senkrechten zu seiner relativen Bewegungsrichtung. Der sogenannte c_w -Wert oder Strömungswiderstandskoeffizient bezieht sich auf das Objekt, und beschreibt dessen Aerodynamik. Zu beachten ist, wenn es um die Betrachtung der Reibung geht, dass die Verbindungsstange, die in der Realität notwendig ist, um die Kugel in ihrer Kreisbahn zu halten, im Modell nur eine unsichtbare Kraft ist, beziehungsweise als unendlich dünne Stange implementiert ist. Daraus ergibt sich, dass jegliche Reibung, die zwischen Stange und Medium oder an einer Verankerung der Stange am Koordinatenursprung entsteht vernachlässigt wird. Sei r der Radius der Pendelkugel, werden der Betrag des Strömungswiderstands und die dabei relevanten Komponenten folgendermaßen berechnet.

$$A = r^2 \cdot \pi$$

$$F_{fr} = \frac{1}{2} \cdot r^2 \pi \cdot c_w \cdot \rho \cdot v^2$$

Nachdem der Betrag des Strömungswiderstands nun berechnet ist, gilt es jetzt den vollständigen Vektor zu bilden. Da die Kraft, die den Strömungswiderstand darstellt, immer entgegengesetzt der Bewegungsrichtung wirkt, ist es zunächst notwendig den Einheitsvektor der Pendelgeschwindigkeit \vec{v} zu ermitteln, was durch das Dividieren derselben durch ihren Betrag erreicht werden kann.

$$\vec{v}_0 = \frac{\vec{v}}{v}$$

Nun kann man den Betrag des Strömungswiderstands mit dem negativen Einheitsvektor der Relativgeschwindigkeit multiplizieren und erhält dann die Formel für den Kraftvektor \vec{F}_{fr} .

$$\vec{F}_{fr} = \frac{1}{2} \cdot r^2 \pi \cdot c_w \cdot \rho \cdot v^2 \cdot (-\vec{v}_0)$$

$$\vec{F}_{fr} = \frac{1}{2} \cdot r^2 \pi \cdot c_w \cdot \rho \cdot v \cdot (-\vec{v}) \quad (5)$$

2.3 Die Coulombkräfte

Wie bereits angemerkt, werden die Magnete im Modell wie elektrisch geladene Kugeln behandelt. Das bietet den Vorteil, dass die Berechnung der dadurch entstehenden Kräfte deutlich erleichtert wird. Es gilt nämlich ohne Einschränkung das Superpositionsprinzip, das heißt, wenn zwei Fixmagnete am Boden eine Wechselwirkung mit dem Pendelmagneten eingehen, dürfen die Vektoren beider Kräfte vektoriell addiert werden. Des weiteren kann die Berechnung des Betrags der Coulombkraft zwischen zwei kugelförmigen Ladungsträgern $L1$ und $L2$ über folgende Formel vollzogen werden, wenn Q_{L1} und Q_{L2} deren Ladungen sind.

$$F_c = \left| \frac{1}{4\pi \cdot \epsilon_0 \cdot \epsilon_r} \cdot \frac{Q_{L1} \cdot Q_{L2}}{L1L2^2} \right|$$

Da sich gleiche Ladungen immer abstoßen und ungleiche sich anziehen, ergäbe sich, würden keine Betragsstriche verwendet, für den Betrag der Ladung bei einer anziehenden Wirkung ein negativer Wert. Diese Tatsache klingt zwar in sich widersprüchlich, heißt aber nur, dass sich im Betrag schon eine Richtung befindet. Um nun einen korrekt gerichteten Vektor zu erhalten, darf für die Berechnung des Vektors \vec{F}_{cL1L2} nicht der Betrag mit dem Einheitsvektor $\frac{L1L2_0}{L1L2^2}$ multipliziert werden. Bei einer Anziehung würde sonst der negative Betragswert mit der korrekten Vektorrichtung multipliziert, was in einem Vektor mit genau entgegengesetzter Richtung resultieren würde. Stattdessen muss der Einheitsvektor $\frac{L2L1_0}{L2L1^2}$ verwendet werden. Es ergibt sich somit zunächst folgende Formel.

$$\begin{aligned} \vec{F}_{cL1L2} &= \frac{1}{4\pi \cdot \epsilon_0 \cdot \epsilon_r} \cdot \frac{Q_{L1} \cdot Q_{L2}}{L1L2^2} \cdot \frac{L2L1_0}{L2L1^2} \\ &= \frac{Q_{L1} \cdot Q_{L2}}{4\pi \cdot \epsilon_0 \cdot \epsilon_r} \cdot \frac{1}{L2L1^2} \cdot \frac{L2L1_0}{L2L1} \\ &= \frac{Q_{L1} \cdot Q_{L2}}{4\pi \cdot \epsilon_0 \cdot \epsilon_r} \cdot \frac{L2L1_0}{L2L1^3} \end{aligned}$$

Betrachtet man nun die Kraft \vec{F}_{cPA} , die der Pendelmagnet P durch den Ablenkungsmagneten A erfährt, und die Kraft \vec{F}_{cPB} , mit der B den Pendelmagneten ablenkt, so ergeben sich folgende Formeln, aus denen schließlich die gesamte auf P einwirkende Coulombkraft \vec{F}_c errechnet werden kann.

$$\begin{aligned} \vec{F}_{cPA} &= \frac{Q_P \cdot Q_A}{4\pi \cdot \epsilon_0 \cdot \epsilon_r} \cdot \frac{\vec{AP}}{AP^3} \\ \vec{F}_{cPB} &= \frac{Q_P \cdot Q_B}{4\pi \cdot \epsilon_0 \cdot \epsilon_r} \cdot \frac{\vec{BP}}{BP^3} \\ \vec{F}_c &= \vec{F}_{cPA} + \vec{F}_{cPB} \\ &= \frac{Q_P \cdot Q_A}{4\pi \cdot \epsilon_0 \cdot \epsilon_r} \cdot \frac{\vec{AP}}{AP^3} + \frac{Q_P \cdot Q_B}{4\pi \cdot \epsilon_0 \cdot \epsilon_r} \cdot \frac{\vec{BP}}{BP^3} \\ \vec{F}_c &= \frac{Q_P}{4\pi \cdot \epsilon_0 \cdot \epsilon_r} \cdot \left(\frac{Q_A \cdot \vec{AP}}{AP^3} + \frac{Q_B \cdot \vec{BP}}{BP^3} \right) \quad (6) \end{aligned}$$

2.4 Die Zentripetalkraft

Den letzten Einflussfaktor der resultierenden Kraft stellt die Zentripetalkraft dar. Ist m die Masse des Objekts, v der Betrag seiner Geschwindigkeit und r der Radius der vom Objekt durchlaufenen Kreisbahn, so gilt diese Formel für den Betrag der Zentripetalkraft.

$$\begin{aligned} F_{zp} &= m \cdot \frac{v^2}{r} \\ &= m \cdot \frac{v^2}{P0} \end{aligned}$$

Um letztendlich den Vektor der Zentripetalkraft zu erhalten, wird ähnlich wie bei den zuvor behandelten Teilkräften der errechnete Betrag mit dem Einheitsvektor, der die Richtung enthält, multipliziert. Hierfür kann der Einheitsvektor $\frac{P0_0}{P0}$ verwendet werden, da er die imaginäre Stange darstellt, von der die Zentripetalkraft zum Kreismittelpunkt hin aufgebracht wird.

$$\begin{aligned} \vec{F}_{zp} &= m \cdot \frac{v^2}{P0} \cdot \frac{P0_0}{P0} \\ &= m \cdot \frac{v^2}{P0} \cdot \frac{P0}{P0} \\ \vec{F}_{zp} &= mv^2 \cdot \frac{P0}{P0^2} \quad (7) \end{aligned}$$

2.5 Die Kräftezerlegung

Um ermitteln zu können, welche Teilkraft \vec{F}_{geg} die Verbindungsstange aufbringen muss, muss der normal zur Pendelbahn gerichtete Teil \vec{F}_{norm} von \vec{F}_{ges} ermittelt werden. Dieser ist genau der Teil, der von der Stange absorbiert werden muss, um eine Entfernung des Pendelmagneten vom Ankerpunkt zu verhindern. Der andere Teil von \vec{F}_{ges} hingegen, \vec{F}_{tan} ist

tangential zur Pendelbahn ausgerichtet und stellt somit alle Kräfte dar, die die Auslenkung des Pendels beeinflussen. Abbildung 2 zeigt alle dabei relevanten Kräfte im Modell.




Abbildung 2: Skizze der Kräftezerlegung

Normalkraft: Zur Bestimmung von \vec{F}_{norm} muss eine orthogonale Zerlegung am Vektor \vec{F}_{ges} durchgeführt werden. Dabei gilt, wie in [2] genauer erklärt wird, dass die Komponente \vec{a}_b eines Vektors \vec{a} mit folgender Formel bestimmt werden kann, wenn \vec{b} die Richtung von \vec{a}_b vorgibt.

$$\vec{a}_b = \frac{\vec{a} \circ \vec{b}}{b^2} \cdot \vec{b}$$

Wendet man dies auf die Kräftezerlegung an \vec{F}_{ges} an, so ergibt sich diese Formel für die Kraft \vec{F}_{norm} , da sie die selbe Richtung wie die Verbindungsstange, also der Vektor \vec{OP} , hat.

$$\vec{F}_{norm} = \frac{\vec{F}_{ges} \circ \vec{OP}}{OP^2} \cdot \vec{OP} \quad (8)$$

Tangentialkraft: Wie am Anfang des Abschnitts 2 bereits erläutert wurde, ist die Tangentialkraft \vec{F}_{tan} der rechtwinklig zur Normalkraft ausgerichtete Teil der Kraft \vec{F}_{ges} . Sie ist also, wie der Name impliziert, tangential zur Bahn des Pendels ausgerichtet und stellt somit den Antrieb für jegliche Richtungsänderung auf der Kreisbahn des Pendelmagneten dar. Da \vec{F}_{ges} und \vec{F}_{norm} bereits bekannt sind, kann \vec{F}_{tan} durch eine Subtraktion errechnet werden.

$$\vec{F}_{tan} = \vec{F}_{ges} - \vec{F}_{norm} \quad (9)$$

Resultierende Kraft: Zur Ermittlung der resultierenden Kraft \vec{F}_{res} müssen sämtliche Kräfte, also die Gesamtheit der äußeren Einwirkungen \vec{F}_{ges} und die Zugkraft der Verbindungsstange \vec{F}_{zug} , addiert werden.

$$\vec{F}_{res} = \vec{F}_{ges} + \vec{F}_{zug}$$

Setzt man hier zunächst die Formeln (2) und (3) ein und ersetzt anschließend \vec{F}_{tan} durch die zuvor dafür aufgestellte Formel (9), so ergibt sich folgende Umformung:

$$\begin{aligned} \vec{F}_{res} &= \vec{F}_{tan} + \vec{F}_{norm} + \vec{F}_{zp} - \vec{F}_{norm} \\ &= \vec{F}_{tan} + \vec{F}_{zp} \\ &= \vec{F}_{ges} - \vec{F}_{norm} + \vec{F}_{zp} \end{aligned}$$

Ersetzt man schließlich die Normalkraft durch die Formel (8), so sind auch alle abstrakteren Größen eliminiert.

$$\vec{F}_{res} = \vec{F}_{ges} - \frac{\vec{F}_{ges} \circ \vec{OP}}{OP^2} \cdot \vec{OP} + \vec{F}_{zp} \quad (10)$$

2.6 Die Differentialgleichungen

Um aus der zuvor bestimmten resultierenden Kraft die Änderung von Geschwindigkeit und Position zu ermitteln, sind nur wenige Formeln nötig. Zunächst kann die auf den Pendelmagneten wirkende Beschleunigung gemäß dem 2. Newtonschen Gesetz über diese Formel bestimmt werden:

$$\vec{a} = \frac{\vec{F}}{m} \quad (11)$$

Für die Geschwindigkeitsänderung \vec{v} und die Positionsänderung \vec{p} gelten zudem diese Formeln:

$$\vec{v} = \vec{a} \quad (12)$$

$$\vec{p} = \vec{v} \quad (13)$$

3 ERSTELLUNG DER MODELLBESCHREIBUNG

Da bei der Umsetzung eines Magnetpendels zu MDL-Code keine mobilen Komponenten benötigt werden, reicht eine einzige Basiskomponente zur Umsetzung dieses Projekts aus. Deren Programmcode ist im Codebeispiel 4 zu finden.

3.1 Die Umsetzung der Vektoren

Simplex3 bietet keine Unterstützung von vektoriellen Variablen, was dazu führt, dass jeder Vektor aus einzelnen skalaren Variablen für jede Dimension modelliert werden muss. Wie im Codebeispiel 1 zu sehen ist, sind diese Einzelvariablen in eine Kommentarstruktur eingefasst, um ihre Zusammengehörigkeit zu verdeutlichen. Die Benennung der Einzelvariablen eines Vektors unterliegt folgendem Muster: `<Bezugsobjekt>_<Objektgroesse>_<Betragsrichtung>`. Im Beispiel handelt es sich also um den Geschwindigkeitsvektor des Pendelmagneten. Die erste Variable stellt dabei dessen Betrag in x-Richtung dar.

3.2 Die Umsetzung physikalischer Einheiten

Die Syntax der Model Description Language von Simplex3 bietet die Möglichkeit, Variablen Einheiten zuzuordnen, sodass automatisch zwischen unterschiedlichen Maßeinheiten umgerechnet werden kann. Zudem bietet diese Möglichkeit den Vorteil, dass eventuell auftretende Fehler in Formeln schneller erkannt werden können, wenn sich eine Inkompatibilität der Einheiten ergibt, weil dies als Fehler ausgegeben wird. Jedoch werden durch die Angabe der Einheiten die im Code vorkommenden Formeln deutlich länger und vor allem unlesbarer, da Simplex3 maximal 80 Zeichen pro Codezeile zulässt. Bis auf Zeitvariablen, die in Sekunden angegeben werden, da die MDL-Syntax bei diesen eine Einheit verlangt, haben deshalb alle Variablen die Einheit 1. Dies erfordert die Konvention, dass alle Variablen in ihrer SI-Basiseinheit geführt werden, welche als Hilfestellung für jede Variable angegeben ist. Zu finden ist die Angabe im Kommentarbereich hinter der zugehörigen Variable.

3.3 Die Abbruchbedingung

Um in Experimenten mit einer hohen Anzahl an Modelldurchläufen Rechenzeit einzusparen, ist das Modell mit einer Abbruchbedingung ausgestattet, welche im Codebeispiel 2 abgebildet ist. Die Abbruchbedingung beendet den Modelllauf, sobald sich der Pendelmagnet über eine gewisse Dauer im unmittelbaren Umfeld eines der Fixmagnete aufgehalten hat. Konkret heißt das, dass sich der Pendelmagnet über eine Zeit von fünf Sekunden nicht weiter in y-Richtung von einem der Fixmagnete entfernen darf, als es in der Variable `Accuracy`,

die die Genauigkeit der Abbruchüberprüfung angibt, festgelegt ist. Dabei ist nur die y-Richtung von Belang, da die Fixmagnete in x-Richtung ohnehin auf der Ruheposition des Pendelmagneten liegen. Wird der Lauf abgebrochen, so hält die Abbruchbedingung sein Ergebnis, also die Endposition des Pendelmagneten in der Variable `Result` fest. Dabei steht der Wert 1 dafür, dass sich der Pendelmagnet über dem Fixmagneten *A* eingependelt hat, während 2 das Äquivalent für Magnet *B* darstellt. Endet das Modell nicht durch das Eintreten der Abbruchbedingung, sondern weil die im Experiment festgelegte Maximallaufzeit verstreicht, so bleibt die Variable `Result` auf dem Initialwert 0. Dieser Fall tritt ein, wenn die Startposition des Pendelmagneten direkt zwischen den Fixmagneten, also auf der Mittelsenkrechte der Strecke $[AB]$, liegt.

Umgesetzt ist diese Abbruchbedingung über die in Simplex-MDL verfügbaren ON-Events. Diese werden nur dann ausgelöst, wenn die nach dem ON gegebene Bedingung vom Zustand `false` auf `true` wechselt, jedoch nicht, wenn im vorherigen Rechenzyklus die Bedingung bereits erfüllt war. In Zeile sieben des Codebeispiels 2 wird die Hilfsvariable `TStop` dann auf die aktuelle Modellzeit `T` gesetzt, wenn der Pendelmagnet gerade in das unmittelbare Umfeld eines der Fixmagneten eintritt. In den folgenden ON-Events wird für jeden der Fixmagnete überprüft, ob nach dem Ablaufen der fünf Sekunden seit dem Setzen der Variable `TStop` die Nähe zum Pendelmagnet noch vorhanden ist. Ist dies der Fall, so wird das Ergebnis des Modelllaufs entsprechend gesetzt und die Simulation beendet.

Wäre der Pendelmagnet vor dem Verstreichen der fünf Sekunden aus der Umgebung des Fixmagneten aus- und wieder in die Umgebung eines der Fixmagneten eingetreten, so wäre beim Wiedereintritt die Variable `TStop` neu gesetzt worden, sodass die unteren beiden Abfragen nicht ausgelöst würden, obwohl die ortsspezifische Bedingung erfüllt ist, da die zur Erfüllung der zeitspezifischen Anforderung nötige Simulationszeit $T > TStop + 5[s]$ noch nicht erreicht ist.

3.4 Der Aufbau des MDL-Codes

Variablendeklaration: Zur Umsetzung der in Abschnitt 2 hergeleiteten Modelldynamik sind in Simplex-MDL, der Modellbeschreibungssprache von Simplex3, mehrere Typen an Variablen notwendig, welche im Teilbereich


```

#--- Start --- Vektor p_v -----
p_v_x (REAL) := 0,           #x Komponente der Geschwindigkeit von P [m/s]
p_v_y (REAL) := 0,           #y Komponente der Geschwindigkeit von P [m/s]
p_v_z (REAL) := 0           #z Komponente der Geschwindigkeit von P [m/s]
#--- Ende --- Vektor p_v -----

```

Programmcode 1: Modellierung des Geschwindigkeitsvektors

```

#Abbruchueberpruefung

#Setze Timeout bei Naehe zum Magneten
ON ^((p_p_y > a_p_y - Accuracy) AND (p_p_y < a_p_y + Accuracy)) OR
((p_p_y > b_p_y - Accuracy) AND (p_p_y < b_p_y + Accuracy))^
DO
    TStop^ := T;
END
#Ueberpruefe nach dem Timeout, ob Naehe noch vorhanden ist.
ON ^(TStop + 5[s] < T) AND ((p_p_y > a_p_y - Accuracy) AND
(p_p_y < a_p_y + Accuracy))^
DO
    Result^ := 1;
    SIGNAL STOP;
END
ON ^(TStop + 5[s] < T) AND ((p_p_y > b_p_y - Accuracy) AND
(p_p_y < b_p_y + Accuracy))^
DO
    Result^ := 2;
    SIGNAL STOP;
END

```

Programmcode 2: Implementierung der Abbruchbedingung

DECLARATION OF ELEMENTS deklariert werden. Sämtliche Natur- und Modellkonstanten, sowie die Positionen der unbeweglichen Fixmagneten A und B werden als Konstanten im entsprechenden Bereich CONSTANTS definiert. Die Hilfsvariablen der Abbruchbedingung TStop und Result sowie Position und Geschwindigkeit des Pendelmagneten sind als Zustandsvariablen unter STATE VARIABLES gelistet. Dabei haben die beiden Letzteren den Zusatz CONTINUOUS, sodass sie über die im Teilabschnitt 2.6 genannten Differentialgleichungen verändert werden können. Im Gegensatz dazu stehen die Variablen für die Kraftvektoren, die fortlaufend über eine direkte Wertzuweisung verändert werden. Dies verlangt nach einer Deklaration im Bereich der DEPENDENT VARIABLES im Abschnitt CONTINUOUS. Die Teilkraft \vec{F}_{ges} ist neben der resultierend auf den Pendelmagneten wirkenden Kraft \vec{F}_{res} der einzige Kraftvektor,

der im Modellcode als Variable abgespeichert wird, während alle anderen Teilkräfte ohne Zwischenspeicherung in der Wertzuweisung berechnet werden. Dieses Vorgehen ist ressourcensparend, da nur die Teilkraft \vec{F}_{ges} mehrfach in der Berechnung der resultierenden Kraft \vec{F}_{res} vorkommt. Desweiteren muss keine der Teilkräfte in nachfolgenden Experimenten observiert werden, was ohne das Abspeichern dieser nicht möglich wäre.

Dynamik: Die Modelldynamik im Teilbereich DYNAMIK BEHAVIOUR des MDL-Codes enthält neben der Abbruchbedingung zunächst die Berechnung der Kraftvektoren \vec{F}_{ges} und \vec{F}_{res} mithilfe der vollständig aufgelösten Formeln (1) und (10). Im Abschnitt DIFFERENTIAL EQUATIONS folgt schließlich die Implementierung der Differentialgleichungen für Geschwindigkeit (12) und Ort (13) in Kombination mit der Formel (11) für die Berechnung der Beschleunigung.

4 EXPERIMENTELLE BESTIMMUNG DES KAUSALITÄTSVERHALTENS

Wie in Abschnitt 1 erwähnt, dient das Magnetpendel oft zur Erläuterung von schwacher Kausalität, da es bei minimaler Änderung der Position beim Loslassen des Pendels ein komplett anderes Resultat liefert. Das Verhalten der Kausalität bei Veränderung bestimmter Einflussfaktoren wird im Folgenden qualitativ bestimmt.

4.1 Die Bestimmung der Kausalität

Um die Kausalität zu bestimmen, werden sogenannte Fraktalbilder des Magnetpendels benötigt. Diese stellen eine eingefärbte Draufsicht des Pendels dar, wobei jedem Magneten eine Farbe zugewiesen wird. Jeder Punkt des Bildes ist in der Farbe des Magneten eingefärbt, über dem das Pendel zur Ruhe kommt, wenn der Pendelmagnet am selbigen Punkt losgelassen wird. Dabei gilt für die Geschwindigkeit des Pendelmagneten $\vec{v} = \vec{0}$. Solch ein Fraktalbild wird durch das Abtasten jedes Punktes in einem gewissen Bereich des Pendels erzeugt, wie es im Codebeispiel 3 im Anhang zu sehen ist. Durch zwei verschachtelte FOR-LOOPS, die jeweils eine Zählervariable mitführen, wird das Feld vom Ursprung in x- und y-Richtung je einen halben Meter weit abgetastet. Diese Zählervariablen stellen einen Integer-Wert dar, der von 0 bis 50 wandert. Dies entspricht dem Abtasten des ersten Quadranten des hier zweidimensional betrachteten Koordinatensystems bei einer Schrittweite von einem Zentimeter. Es wird hier nur der erste Quadrant abgetastet, da das Modell symmetrisch aufgebaut ist und somit aus einem der Quadranten die Restlichen durch Spiegelung und Farbinvertierung generiert werden können. So kann nach einmaliger Bestätigung der Symmetrie die benötigte Rechenzeit auf ein Viertel reduziert werden.

Für jeden Punkt, also innerhalb der FOR-LOOPS, wird ein Durchlauf der in Abschnitt 3 beschriebenen Modellbeschreibung gestartet. Dabei wird aus den Zählervariablen und der Länge der Verbindungsstange des Pendels, die in der Variable `l` gespeichert ist, die aktuelle Startposition des Pendelmagneten errechnet. Nach dem Ende des Simulationslaufs wird die Modellvariable `Result` abgefragt, welche, wie in Teilabschnitt 3.3 beschrieben, den Endzustand des Pendels enthält. Diese wird dann zusammen mit den x- und y-Koordinaten

des Laufs in der Konsole ausgegeben. Danach wird durch die FOR-LOOPS die Position verändert und ein neuer Lauf gestartet.

Aus der so erzeugten Konsolenausgabe kann mithilfe des Programms [3] ein Fraktalbild generiert werden. Dem Magneten *B* bzw. der Zahl 2 ist dabei die Farbe blau und dem Magneten *A*, also der Zahl 1, die Farbe rot zugeordnet. Liefert der Lauf kein Ergebnis, also den Initialwert 0, so wird der entsprechende Punkt weiß gekennzeichnet.

Die Stärke der Kausalität lässt sich nun anhand der Anzahl und Größe der auf dem Fraktalbild sichtbaren, zusammenhängenden, einfarbigen Felder bestimmen. Besteht ein Fraktalbild aus vielen winzigen Feldern, weist dies auf schwache Kausalität hin, während große Felder auf starke Kausalität hindeuten.

4.2 Die Wertfindung

Da es für ein Magnetpendel keinerlei Normen gibt, beruhen sämtliche Modellkonstanten wie die Positionen der Fixmagnete oder Größe und Masse des Pendelmagneten auf Schätzungen. Lediglich die Ladungen der elektrostatischen Kugeln *A* und *B*, sowie des Pendelmagneten bilden dabei eine Ausnahme. Diese sind so gewählt, dass bei jeder im Rahmen der Abtastgenauigkeit erreichbaren Startposition die resultierende Coulombkraft \vec{F}_c die auf den Pendelmagneten wirkende Erdanziehungskraft \vec{F}_g überwinden kann, sodass lediglich Punkte, die auf der Mittelsenkrechten der Strecke $[AB]$ liegen, kein Resultat liefern, also weiß eingefärbt werden.

4.3 Die Untersuchung des Kausalitätsverhaltens

Durch die Durchführung des zuvor beschriebenen Experiments unter verschiedenen Umwelteinflüssen entstanden die in den Abbildungen 3, 4, 5 und 6 dargestellten Fraktalbildreihen. Jede Bildreihe stellt eine Versuchsreihe dar, wobei stets nur ein Wert verändert wird. Dazu wird vor jedem Experimentlauf der Initialwert der entsprechenden Konstante im MDL-Code verändert. Das Teilbild (a) der Fraktalbildreihe zeigt stets das Fraktal mit der schwächsten Kausalität, das Teilbild (c) das Fraktal mit der stärksten Kausalität und das Teilbild (b) den über alle Versuchsreihen gleichbleibenden Vergleichswert. Letzterer ist so gewählt, dass bei der gewählten Auflösung Strukturen im Fraktalbild erkennbar sind. Dadurch bleibt Raum für Veränderungen der

Kausalität in beide Richtungen: Entweder hin zu reinem Bildrauschen, also schwacher Kausalität oder komplett einfarbigen Bildhälften und somit vollständiger Determiniertheit.

Dichte des Mediums: Das Fraktal (a) in Abbildung 3 zeigt klar, wieso das Magnetpendel oft als Beispiel für schwache Kausalität verwendet wird. Bei dieser Versuchsreihe wurde lediglich die Stärke des Strömungswiderstands, oder genauer die Dichte des Mediums, durch das sich der Pendelmagnet bewegt, verändert, sodass das Experiment stets unter erdähnlichen Bedingungen stattfand. In Luft, wo das Pendel üblicherweise anzutreffen ist, sind nur wenige Strukturen erkennbar, die jedoch sehr klein sind. Sie könnten deshalb auch auf Zufall beruhen, da bei der gewählten Abtastrate von einem Zentimeter eine hohe Wahrscheinlichkeit existiert, dass zwischen zwei abgetasteten Punkten mit gleichem Ergebnis ein Punkt mit gegenteiligem Ergebnis unentdeckt bleibt. Je größer aber die zusammenhängende Fläche ist, desto kleiner wird die Wahrscheinlichkeit, dass die Fläche auf das soeben beschriebene Phänomen zurückzuführen ist.

Tauscht man hingegen das Medium gegen eines mit höherer Dichte, so gibt das Pendel seinen chaotischen Charakter schnell auf. Teilbild (b) zeigt den Versuch in Wolframhexafluorid. Dabei handelt es sich um ein Gas mit der Dichte $\rho = 12.7 \frac{\text{kg}}{\text{m}^3}$, was ca. dem Zehnfachen der Dichte von Luft entspricht. Während sich dort schon größere Strukturen erkennen lassen, welche erste Vorhersagen bei einem mit der Hand durchgeführten Pendellauf ermöglichen würden, ist das Pendel in Wasser komplett determiniert, wie im Fraktal (c) zu erkennen ist. Dies lässt sich dadurch erklären, dass bei der mit $1000 \frac{\text{kg}}{\text{m}^3}$ überaus hohen Dichte von Wasser der Pendelmagnet einem so starken Strömungswiderstand ausgesetzt ist, dass er nicht genug Geschwindigkeit aufbauen kann, um über einen Fixmagneten hinwegzuschwingen. Demnach kommt das Pendel beim nächstgelegenen Fixmagneten zum Stehen. Bei diesem Versuch müsste allerdings der mit dem Gesetz von Stokes definierte Strömungswiderstand für Bewegungen in Flüssigkeiten verwendet und die Auftriebskraft, die der Pendelmagnet im Wasser erfährt, berücksichtigt werden. Jedoch würde dies in noch viel höheren rücktreibenden Kräften resultieren, sodass der Effekt der Determiniertheit, der bereits maximal ist, nur noch verstärkt würde.

Masse des Pendelmagneten: In der Versuchsreihe aus Abbildung 4 wurde die Masse des

Pendelmagneten variiert. Dies nimmt Einfluss auf dessen Trägheit. Hier zeigt sich, welchen großen Einfluss die Ausführung des Pendels auf dessen Kausalität haben kann. Während die Ausführung (c) mit leichtem Pendel stark kausal ist, weist das Fraktalbild (a) mit schwerem Pendel in großen Teilen starke Ähnlichkeit mit dem Fraktalbild (a) aus Abbildung 3 auf, obwohl der Versuch des ersteren in Wolframhexafluorid und nicht in Luft durchgeführt wurde.

Stärke der Magnete: Selbiges zeigt sich in den Fraktalbildern aus Abbildung 5, bei denen die Ladung der Magnete bzw. elektrostatisch geladenen Kugeln verändert wurde. Der zugehörige Betrag der Ladung ist unter jedem Teilbild angegeben. Dabei bleibt festzuhalten, dass die Determiniertheit mit den magnetischen Anziehungskräften steigt.

Stärke der Fallbeschleunigung: Im Kontrast dazu stehen die Versuche aus Abbildung 6. Dort zeigt sich, dass die Kausalität mit zunehmender Fallbeschleunigung von der in Schwerelosigkeit ($0 \frac{\text{m}}{\text{s}^2}$) bis hin zu der des Jupiters ($24.8 \frac{\text{m}}{\text{s}^2}$) abnimmt.

Erklärung der Beobachtungen: Relevant für die Kausalität des Pendels ist die Dauer, die der Pendelmagnet benötigt, um so viel Geschwindigkeit abzubauen, dass er von einem Fixmagneten festgehalten werden kann. Hat man zwei Läufe, in denen der Pendelmagnet an nahezu der selben Startposition losgelassen wird, so divergieren die Bahnen dieser Pendelmagnete umso stärker, je länger die Pendel bereits schwingen. Dies ist darauf zurückzuführen, dass die Ablenkung durch die Fixmagnete abhängig von deren Abstand zum Pendelmagneten ist und somit eine Differenz zwischen den Bahnen auch eine Differenz in den Kräften, die die Bahnen beeinflussen, zur Folge hat. Dieses Phänomen ist auch bei normalen Magnetpendeln bei denen Magnete anstelle der geladenen Kugeln verwendet werden zu erwarten, da auch deren Anziehungskräfte zumindest näherungsweise abhängig vom Abstand sind. Besonders stark nimmt die Anziehung dann ab, wenn die Pole der Magneten nicht mehr aufeinandergerichtet sind. Befindet sich der Pendelmagnet also nur minimal neben einem Fixmagneten, so sinkt die Anziehungskraft zwischen den Magneten bereits drastisch. Die auftretenden Differenzen zwischen zwei solchen Bahnen, wie sie zuvor beschrieben wurden, können somit noch stärker ausfallen, als in dem hier implementierten Pendel.

Gegen Ende eines Laufes ist der Pendelmagnet

sehr langsam, wobei er meist zwischen den Fixmagneten hin und her schwingt indem er diese halb umrundet, sodass seine Bahn einer Acht gleicht. In dieser Phase kann eine minimale Differenz in Position oder Geschwindigkeit darüber entscheiden, ob der Pendelmagnet seine bisherige Bahn beibehält oder nur noch in einem Kreis um einen der Fixmagnete schwingt. Genauso kann der Pendelmagnet aufgrund solch einer kleinen Differenz aus seiner Bahn um einen der Fixmagnete ausbrechen und wieder zurück in die Bahn um beide Fixmagnete wechseln. So können zwei Läufe, mit nahezu gleichen Anfangsbedingungen ein komplett gegenteiliges Ergebnis liefern. Da die dafür nötige Differenz umso leichter zu erreichen ist, je länger die elektrostatischen Anziehungskräfte wirken konnten, nimmt die Kausalität bzw. Determiniertheit des Magnetpendels mit jeder Sekunde, die es schwingt, ab. Dies kann anhand der Beobachtungen bestätigt werden. Mit zunehmender Dichte des Mediums steigen die rücktreibenden Kräfte, der Pendelmagnet wird schneller gebremst und schwingt somit nur für eine kürzere Zeit. Steigt die Masse und somit die Trägheit des Pendelmagneten, so benötigt es bei gleicher Kraft mehr Zeit diesen zu bremsen. Mit der Ladung der elektrostatischen Kugeln steigt die Kraft, mit der der Pendelmagnet in der Kreisbahn um einen der Fixmagnete gehalten werden kann. Dies führt dazu, dass dieser nicht so stark abgebremst werden muss, um einen vollständig determinierten Zustand zu erreichen. Schließlich führt eine erhöhte Fallbeschleunigung g laut der Formel $E_{pot} = m \cdot g \cdot \Delta h$ zu einer höheren potentiellen Energie, die der Pendelmagnet zu Beginn des Laufs bei konstanter Höhendifferenz Δh aufweist. Diese wird im Zuge jedes Pendelschwunges in kinetische Energie also Geschwindigkeit umgewandelt. Da diese somit ebenso steigt, wird mehr Zeit benötigt, um den Pendelmagneten zu bremsen.

Die Dauer des Pendelvorgangs ist somit ausschlaggebend für die Stärke der Kausalität eines Magnetpendels. Je länger ein Pendel aufgrund seines Aufbaus und der gegebenen Umwelteinflüsse durchschnittlich schwingt, desto chaotischer ist es.

LITERATUR

- [1] C. Cornilsen. „Das Magnetpendel als Weg zum Verständnis von Grundideen der Chaostheorie“. In: *Staatsexamenarbeit für das Lehramt an höheren Schulen. IPN an der Universität Kiel* (1993).
- [2] K. Meyberg und P. Vachenauer. *Höhere Mathematik 1: Differential- und Integralrechnung Vektor- und Matrizenrechnung*. Springer-Lehrbuch. Springer Berlin Heidelberg, 2013, S. 26. ISBN: 9783642972140. URL: <https://books.google.de/books?id=rMqhBgAAQBAJ> (besucht am 30.10.2017).
- [3] M. Obermeier. *extendedPlotter*. Version 1.4. 20. Aug. 2017. URL: <https://github.com/theMomax/extendedPlotter> (besucht am 30.10.2017).
- [4] B. Schmidt. *The Art of Modelling and Simulation: Introduction to the Simulation System Simplex 3*. Society for Modeling und Simulation International, 2001. ISBN: 9781565552289. URL: <https://books.google.de/books?id=Y2hfXwAACAAJ> (besucht am 30.10.2017).

ANHANG

```

EXPERIMENT DefaultExp

DECLARATION OF VARIABLES
  runCount (INT) := 1,      # Anzahl der Modelllaeufe [1]
  x (REAL) := 0,           # x-Koordinate des aktuellen Punkts [m]
  y (REAL) := 0,           # y-Koordinate des aktuellen Punkts [m]
  l (REAL) := 1,           # Laenge der Verbindungsstange [m]
  result (INT) := 0        # Ergebnis des Modelllaufs [1]

BODY OF EXPERIMENT

# Generieren eines Modelllaufs
CREXP Magnetpendel;
CRRUN Run, Magnetpendel;
SETCTRL StepSize, 0.01;
SETCTRL StepMax, 0.01;

# Veraenderung des aktuellen Punkts
FOR i FROM 0 TO 50 REPEAT
  x := i / 100;
  FOR j FROM 0 TO 50 REPEAT
    y := j / 100;
    # Zuruecksetzen des letzten Modelllaufs
    IF runCount > 1 DO
      SELRUN Run;
      RESETRUN Run;
    END
    # Zuweisung der Startposition
    <p_p_x> := x;
    <p_p_y> := y;
    <p_p_z> := -SQRT(SQE(l) - SQE(y) - SQE(x));
    # Durchfuehrung des Modelllaufs
    SIMULATE TO, 600;
    # Abfrage und Ausgabe des Ergebnisses
    result := <Result>;
    DISPLAY("$Data / %lg / %lg / %d \n", x, y, result);
    runCount := runCount + 1;
  END LOOP
END LOOP
END OF DefaultExp

```

Programmcode 3: EDL-Experimentbeschreibungscod

```

BASIC COMPONENT Magnetpendel

USE OF UNITS
  TIMEUNIT = [s]

DECLARATION OF ELEMENTS

CONSTANTS

```

```

#Naturkonstanten
g (REAL) := 9.81,           #Erdbeschleunigung [m/s^2]
pi (REAL) := 3.1415,       #Pi [1]
e0 (REAL) := 8.854E-12,    #epsilon0 [As^2/N * m^2]
eR (REAL) := 1.00059,     #epsilonR [1]
cw (REAL) := 0.45,        #Cw-Wert [1]
rho (REAL) := 1.2041,     #Dichte des Mediums [kg/m^3]

#Modellkonstanten
Accuracy (REAL) := 0.04,   #Genauigkeit der Abbruchbedingung

#konstante Eigenschaften von P
p_m (REAL) := 0.02,        #Masse des Pendelmagneten P [kg]
p_r (REAL) := 0.05,        #Radius von P [m]
p_q (REAL) := -4E-7,       #Ladung von P [As]

#Elektrostatische Kugel A:
a_q (REAL) := 4E-7,        #Ladung von A [As]
#--- Start --- Vektor a_p -----
a_p_x (REAL) := 0,         #x Komponente der Position von A [m]
a_p_y (REAL) := -0.05,     #y Komponente der Position von A [m]
a_p_z (REAL) := -1.05,     #z Komponente der Position von A [m]
#--- Ende --- Vektor a_p -----

#Elektrostatische Kugel B:
b_q (REAL) := 4E-7,        #Ladung von A [As]
#--- Start --- Vektor b_p -----
b_p_x (REAL) := 0,         #x Komponente der Position von B [m]
b_p_y (REAL) := 0.05,      #y Komponente der Position von B [m]
b_p_z (REAL) := -1.05      #z Komponente der Position von B [m]
#--- Ende --- Vektor b_p -----

STATE VARIABLES

TStop (REAL[s]) := 0[s],   #Markierung fuer Abbruchbedingung [s]
Result (INT) := 0          #Endzustand des Experiments [1]

CONTINUOUS

#Elektrostatische Kugel P:
#--- Start --- Vektor p_p -----
p_p_x (REAL) := 0,         #x Komponente der Position von P [m]
p_p_y (REAL) := 0,         #y Komponente der Position von P [m]
p_p_z (REAL) := -1,        #z Komponente der Position von P [m]
#--- Ende --- Vektor p_p -----
#--- Start --- Vektor p_v -----
p_v_x (REAL) := 0,         #x Komponente der Geschwindigkeit von P [m/s]
p_v_y (REAL) := 0,         #y Komponente der Geschwindigkeit von P [m/s]
p_v_z (REAL) := 0          #z Komponente der Geschwindigkeit von P [m/s]
#--- Ende --- Vektor p_v -----

DEPENDENT VARIABLES

CONTINUOUS

```

```

#--- Start --- Vektor p_fges -----
p_fges_x (REAL) := 0,           #x Komponente von p_fges [N]
p_fges_y (REAL) := 0,           #y Komponente von p_fges [N]
p_fges_z (REAL) := 0,           #z Komponente von p_fges [N]
#--- Ende --- Vektor p_fges -----
#--- Start --- Vektor p_fres -----
p_fres_x (REAL) := 0,           #x Komponente von p_fres [N]
p_fres_y (REAL) := 0,           #y Komponente von p_fres [N]
p_fres_z (REAL) := 0           #z Komponente von p_fres [N]
#--- Ende --- Vektor p_fres -----

```

DYNAMIC BEHAVIOUR

#Abbruchueberpruefung

#Setze Timeout bei Naehe zum Magneten

```

ON ^((p_p_y > a_p_y - Accuracy) AND (p_p_y < a_p_y + Accuracy)) OR
((p_p_y > b_p_y - Accuracy) AND (p_p_y < b_p_y + Accuracy))^

```

DO

TStop^ := T;

END

#Ueberpruefe nach dem Timeout, ob Naehe noch vorhanden ist.

```

ON ^((TStop + 5[s] < T) AND ((p_p_y > a_p_y - Accuracy) AND
(p_p_y < a_p_y + Accuracy))^

```

DO

Result^ := 1;

SIGNAL STOP;

END

```

ON ^((TStop + 5[s] < T) AND ((p_p_y > b_p_y - Accuracy) AND
(p_p_y < b_p_y + Accuracy))^

```

DO

Result^ := 2;

SIGNAL STOP;

END

#Kraefteberechnung

```

#--- Start --- Kraefteberechnung von p_fges -----

```

```

p_fges_x := (p_m * g * 0) + 0.5 * SQE(p_r) * pi * cw * rho *
SQRT((SQE(p_v_x) + SQE(p_v_y) + SQE(p_v_z))) * (-1 * p_v_x) + (((p_q)/
(4 * pi * e0 * eR)) * ((a_q * (p_p_x - a_p_x))/
(POW(SQRT(SQE(p_p_x - a_p_x) + SQE(p_p_y - a_p_y) +
SQE(p_p_z - a_p_z)),3))) + ((b_q * (p_p_x - b_p_x))/
(POW(SQRT(SQE(p_p_x - b_p_x) + SQE(p_p_y - b_p_y) +
SQE(p_p_z - b_p_z)),3))));

```

```

p_fges_y := (p_m * g * 0) + 0.5 * SQE(p_r) * pi * cw * rho *
SQRT(SQE(p_v_x) + SQE(p_v_y) + SQE(p_v_z)) * (-1 * p_v_y) + (((p_q)/
(4 * pi * e0 * eR)) * ((a_q * (p_p_y - a_p_y))/
(POW(SQRT(SQE(p_p_x - a_p_x) + SQE(p_p_y - a_p_y) +
SQE(p_p_z - a_p_z)),3))) + ((b_q * (p_p_y - b_p_y))/
(POW(SQRT(SQE(p_p_x - b_p_x) + SQE(p_p_y - b_p_y) +
SQE(p_p_z - b_p_z)),3))));

```



```

p_fges_z := (p_m * g * (-1)) + 0.5 * SQE(p_r) * pi * cw * rho *
SQRT(SQE(p_v_x) + SQE(p_v_y) + SQE(p_v_z)) * (-1 * p_v_z) + (((p_q) /
(4 * pi * e0 * eR)) * ((a_q * (p_p_z - a_p_z)) /
(POW(SQRT(SQE(p_p_x - a_p_x) + SQE(p_p_y - a_p_y) +
SQE(p_p_z - a_p_z)),3))) + ((b_q * (p_p_z - b_p_z)) /
(POW(SQRT(SQE(p_p_x - b_p_x) + SQE(p_p_y - b_p_y) +
SQE(p_p_z - b_p_z)),3)))));
#--- Ende --- Vektor Kraefteberechnung von p_fges -----

#--- Start --- Vektor Kraefteberechnung von p_fres -----
p_fres_x := p_fges_x - (((p_fges_x * p_p_x + p_fges_y * p_p_y +
p_fges_z * p_p_z) / (SQE(p_p_x) + SQE(p_p_y) + SQE(p_p_z))) * p_p_x) +
(p_m * (SQE(p_v_x) + SQE(p_v_y) + SQE(p_v_z)) *
((-1 * p_p_x) / (SQE(p_p_x) + SQE(p_p_y) + SQE(p_p_z))));

p_fres_y := p_fges_y - (((p_fges_x * p_p_x + p_fges_y * p_p_y +
p_fges_z * p_p_z) / (SQE(p_p_x) + SQE(p_p_y) + SQE(p_p_z))) * p_p_y) +
(p_m * (SQE(p_v_x) + SQE(p_v_y) + SQE(p_v_z)) *
((-1 * p_p_y) / (SQE(p_p_x) + SQE(p_p_y) + SQE(p_p_z))));

p_fres_z := p_fges_z - (((p_fges_x * p_p_x + p_fges_y * p_p_y +
p_fges_z * p_p_z) / (SQE(p_p_x) + SQE(p_p_y) + SQE(p_p_z))) * p_p_z) +
(p_m * (SQE(p_v_x) + SQE(p_v_y) + SQE(p_v_z)) *
((-1 * p_p_z) / (SQE(p_p_x) + SQE(p_p_y) + SQE(p_p_z))));
#--- Ende --- Vektor Kraefteberechnung von p_fres -----

DIFFERENTIAL EQUATIONS

#--- Start --- Geschwindigkeitsberechnung von P -----
p_v_x' := p_fres_x / p_m;
p_v_y' := p_fres_y / p_m;
p_v_z' := p_fres_z / p_m;
#--- Ende --- Geschwindigkeitsberechnung von P -----


#--- Start --- Positionsberechnung von P -----
p_p_x' := p_v_x;
p_p_y' := p_v_y;
p_p_z' := p_v_z;
#--- Ende --- Positionsberechnung von P -----

END



END OF Magnetpendel

```



Programmcode 4: MDL-Modellbeschreibungscodel



(a) Medium: Luft

(a) Pendelmasse: $0.04kg$ 

(b) Medium: Wolframhexafluorid

(b) Pendelmasse: $0.02kg$ 

(c) Medium: Wasser




(c) Pendelmasse: $0.01kg$

Abbildung 3: Fraktalbilder bei veränderter Dichte des Mediums


Abbildung 4: Fraktalbilder bei veränderter Pendelmasse




(a) Magnetladung: $1 \cdot 10^{-7} C$




(a) Standort: Jupiter




(b) Magnetladung: $4 \cdot 10^{-7} C$



(b) Standort: Erde



(c) Magnetladung: $4 \cdot 10^{-6} C$



(c) Standort: Schwerelosigkeit

Abbildung 5: Fraktalbilder bei veränderter Magnetstärke

Abbildung 6: Fraktalbilder bei veränderter Fallbeschleunigung

Erklärung zur Seminararbeit

Hiermit erkläre ich, dass ich die vorliegende Arbeit selbstständig und ohne fremde Hilfe verfasst und keine anderen als die angegebenen Hilfsmittel verwendet habe.

Insbesondere versichere ich, dass ich alle wörtlichen und sinngemäßen Übernahmen aus anderen Werken als solche kenntlich gemacht habe.

Dorfen, den

.....

Unterschrift